```
> restart:
```
We load our package:
```
> with(IntegrableConnections);
```
[*Eigenring*, *Evalm*, *G_matrix*, *G_matrix0*, *HyperexponentialSolutions*, *Id*, *Mapply*, *MatrixColumnPrimpart*, *MatrixColumnPrimpartRat*, *Mexpsolde*, *Mpolsolde*, *Mratsolde*, *Msylvester*, *Mylinsolve*, *NNIexponents*, *PolynomialSolutions*, *RationalSolutions*, *Reduction*, *TestIntegrabilityConditions*, *Trigonalize*, *backsubs*, *candidates*, *col_pval*, *colechelon*, *colval*, *complement*, *convert_back*, *denomatrix*, *denomvector*, *direct_ratsol*, *eval_block_reduced*, *evalam*, *exp_parts*, *formal_reduce*, *gen_sylvester*, *getargs*, *good_form*, *intDiff*, *int_diff_split*, *l_colechelon*, *l_column_rank*, *l_moser_reduce*, *l_qtcd*, *l_super*, *l_super_reduce*, *lc_evala*, *lc_evalam*, *localize*, *log_collect*, *log_free*, *mat_block_reduce*, *mat_change_exp_part*, *mat_check*, *mat_coeff*, *mat_colvaluation*, *mat_convert*, *mat_copy*, *mat_difference*, *mat_eval*, *mat_get_dim*, *mat_get_exp_part*, *mat_get_ext*, *mat_get_indicial_polynomial*, *mat_get_inv_transformation*, *mat_get_order*, *mat_get_ramification*, *mat_get_terms*, *mat_get_transformation*, *mat_get_type*, *mat_lc*, *mat_lead_mon*, *mat_product*, *mat_pval*, *mat_subs*, *mat_sum*, *mat_swap*, *mat_to_simple*, *mat_tools*, *mat_transform*, *mat_val*, *mat_valuation*, *matrix_series*, *matval*, *minimalIntegerRoot*, *mult*, *my_linsolve*, *mydegree*, *myequaind*, *new_matrix*, *newton*, *newton_polygon*, *nonzero*, *normalm*, *pencil_block_reduce*, *pencil_solve*, *pgaussUnimod*, *poly_pval*, *poly_sols*, *poly_val*, *prank*, *pval*, *qtcd*, *ramified_case*, *ramified_reduction*, *randomMatrix*, *randomPoly*, *randomRat*, *ratsuper*, *reduceColRank4*, *reg_sols*, *regular_series*, *remove*, *rowechelon*, *seq_evala*, *sim_sylvester*, *simple_eval*, *simple_sols*, *sortColumns*, *super_form*, *system_expsolde*, *tensor_prod*, *transform*, *truncate*, *utils*, *val*, *val1*, *val_ldegree*, *valuation*, *vectdegree*] **(1)**

The library linalg is also needed.
```
> with(linalg):
```
We load OreModules in order to use the procedure to write a D-finite partial differential system as an integrable connection.
```
> with(OreModules):
```

Consider Example 3.2 in ***Z. Li, F. Schwarz, and S. Tsarev.Factoring systems of pde's with finite-dimensional solution space,***
Journal of Symbolic Computation, 36:443--471, 2003.

We define the OreAlgebra (needed for OreModules):
```
> Alg:=DefineOreAlgebra(diff=[d1,x1],diff=[d2,x2],polynom=[x1,x2],
  comm=[]):
```
We give the equations of the system:
```
> R:=matrix(2,1,[d1^3+(x2^2+6*x1^2-6*x1*x2)/(2*x1^3-x2*x1^2)*d1^2,
  d2^3+(3*x1-2*x2)/(x1^2-x1*x2)*d2^2+(2*x1-x2)/(x1^3-x1^2*x2)*d2]);
```
(2)

$$R := \left[\begin{array}{c} d1^3 + \dfrac{\left(x2^2 + 6\,x1^2 - 6\,x1\,x2\right) d1^2}{2\,x1^3 - x2\,x1^2} \\ d2^3 + \dfrac{(3\,x1 - 2\,x2)\,d2^2}{x1^2 - x1\,x2} + \dfrac{(2\,x1 - x2)\,d2}{x1^3 - x2\,x1^2} \end{array}\right] \tag{2}$$

We write the system as an integrable connection:

```
> C:=OreModules[Connection](R,Alg);
```

$$C := \left[\left[\begin{array}{cccc} 0 & 0 & 1 & 0 \\ 0 & \dfrac{-x1^2 + 3\,x1\,x2 - x2^2}{x1^3 - x2\,x1^2} & 0 & \dfrac{x2}{x1 - x2} \\ 0 & \dfrac{2\,x2\,x1^2 - 5\,x1\,x2^2 + 2\,x2^3}{x1^4 - x1^3\,x2} & 0 & \dfrac{-2\,x1\,x2^2 + x2^3}{x1^3 - x2\,x1^2} \\ 0 & \dfrac{2}{x1^2} & 0 & \dfrac{x2}{x1^2} \end{array}\right], \left[\begin{array}{cccc} 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 \\ 0 & \dfrac{-x1^2 + 3\,x1\,x2 - x2^2}{x1^3 - x2\,x1^2} & 0 & \dfrac{x2}{x1 - x2} \\ 0 & \dfrac{-2\,x1 + x2}{x1^3 - x2\,x1^2} & 0 & \dfrac{-3\,x1 + 2\,x2}{x1^2 - x1\,x2} \end{array}\right]\right] \tag{3}$$

If we want to know the basis of the associated module in which the connection is written, we can use the procedure KBasis of OreModules:

```
> OreModules[KBasis](R,Alg);
```

$$\left[\lambda_1,\ d2\,\lambda_1,\ \lambda_1\,d1,\ d2^2\,\lambda_1\right] \tag{4}$$

We use our procedure for computing hyperexponential solutions of the integrable connection:

```
> HyperexpSols:=HyperexponentialSolutions(C,[x1,x2]);
```

(5)

$$HyperexpSols := \begin{bmatrix} x1 & 1 & -e^{-\frac{x2}{x1}}\, x2^2 & -e^{-\frac{x2}{x1}} \\ 0 & 0 & -\frac{e^{-\frac{x2}{x1}}\, x2\,(2\,x1 - x2)}{x1} & \frac{e^{-\frac{x2}{x1}}}{x1} \\ 1 & 0 & -\frac{e^{-\frac{x2}{x1}}\, x2^3}{x1^2} & -\frac{e^{-\frac{x2}{x1}}\, x2}{x1^2} \\ 0 & 0 & -\frac{e^{-\frac{x2}{x1}}\,(2\,x1^2 - 4\,x1\,x2 + x2^2)}{x1^2} & -\frac{e^{-\frac{x2}{x1}}}{x1^2} \end{bmatrix} \quad \textbf{(5)}$$

The rational solutions are given by the columns of the first matrix. This means the original system admits two non-zero rational solutions
given by 1 and x1. The two other non-zero hyperexponential solutions are given by the second matrix. This means that the original system also admits
exp(-x2/x1) and x2^2*exp(-x2/x1) as hyperexponential solutions.

Consider Example 3.3 in *Z. Li, F. Schwarz, and S. Tsarev.Factoring systems of pde's with finite-dimensional solution space,*
Journal of Symbolic Computation, 36:443--471, 2003.

We define the OreAlgebra (needed for OreModules):

```
> Alg:=DefineOreAlgebra(diff=[d1,x1],diff=[d2,x2],diff=[d3,x3],
  polynom=[x1,x2,x3],comm=[]):
```

We give the equations of the system:

```
> R:=matrix(3,1,[d1^2-x1/(x1-1)*d1+1/(x1-1),d2+x1/(x2*(x1*x2-x2))*d1-
  x1/(x2*(x1*x2-x2)),d3-(2*x1*x3+1/2*x1)/(x1*x3-x3)*d1+(2*x3+1/2*x1)/
  (x1*x3-x3)]);
```

$$R := \begin{bmatrix} d1^2 - \frac{x1\, d1}{x1 - 1} + \frac{1}{x1 - 1} \\ d2 + \frac{x1\, d1}{x2\,(x1\, x2 - x2)} - \frac{x1}{x2\,(x1\, x2 - x2)} \\ d3 - \frac{\left(2\, x1\, x3 + \frac{1}{2}\, x1\right) d1}{x1\, x3 - x3} + \frac{2\, x3 + \frac{1}{2}\, x1}{x1\, x3 - x3} \end{bmatrix} \quad \textbf{(6)}$$

We write the system as an integrable connection:

```
> C:=OreModules[Connection](R,Alg):
```

If we want to know the basis of the associated module in which the connection is written, we can use the procedure KBasis of OreModules:

```
> OreModules[KBasis](R,Alg);
```

$$\left[\lambda_1,\, d3\, \lambda_1\right] \quad \textbf{(7)}$$

We use our procedure for computing hyperexponential solutions of the integrable connection:

```
> HyperexpSols:=HyperexponentialSolutions(C,[x1,x2,x3]);
```

$$HyperexpSols := \begin{bmatrix} -\frac{2\,x1\,e^{-\frac{1}{x2}}}{\sqrt{x3}} & \frac{1}{2}\,e^{x1+2\,x3} \\ \frac{x1\,e^{-\frac{1}{x2}}}{x3^{3/2}} & e^{x1+2\,x3} \end{bmatrix} \quad \textbf{(8)}$$

The system admits two non-zero hyperexponential solutions, namely exp(x1+2*x3) and x1/(sqrt(x3))*exp(-1/x2).

Consider Example 3.4 in *Z. Li, F. Schwarz, and S. Tsarev.Factoring systems of pde's with finite-dimensional solution space,*
Journal of Symbolic Computation, 36:443--471, 2003.

We define the OreAlgebra (needed for OreModules):

```
> Alg:=DefineOreAlgebra(diff=[d1,x1],diff=[d2,x2],diff=[d3,x3],
  polynom=[x1,x2,x3],comm=[]):
```

We give the equations of the system:

```
> v:=x1^2-x1*x3+x2^2+2*x2*x3+x3^2;
```

$$v := x1^2 - x1\,x3 + x2^2 + 2\,x2\,x3 + x3^2 \quad \textbf{(9)}$$

```
> w:=-2*x2^4-8*x3*x2^3-12*x2^2*x3^2-8*x2*x3^3+2*x1*x3^3+2*x1^3*x2-3*
  x1^2*x3^2-4*x1^2*x2*x3-2*x3^4+x1^4+2*x1*x2*x3^2;
```

$$w := -2\,x2^4 - 8\,x3\,x2^3 - 12\,x2^2\,x3^2 - 8\,x2\,x3^3 + 2\,x1\,x3^3 + 2\,x1^3\,x2 - 3\,x1^2\,x3^2 - 4\,x1^2\,x2\,x3 - 2\,x3^4 + x1^4 + 2\,x1\,x2\,x3^2 \quad \textbf{(10)}$$

```
> R:=matrix(4,1,[
  x1*v*d1-x2*v*d2+x1*(2*x1*x2+x1*x3+x2^2)*d3-2*x1*x2-x1*x3+x1^2,
  x1*(x1-x3)*(2*x2-x1+2*x3)*d3+(x2+x3)^2*v*d2^2+2*x1*v*d2+x1*(-2*x2+
  x1-2*x3),
  (x2+x3)^2*v*d2*d3+x1*v*d2+2*x1*(x2+x3)*(x1-x3)*d3-2*x1*(x2+x3),
  (x2+x3)^2*(x1-x3)*v*d3^2+w*d3-2*x1*x2^2-2*x1^2*x2-2*x1*x2*x3-x1^3-
  x3*x1^2
  ]);
```

$$R := [[x1\,(x1^2 - x1\,x3 + x2^2 + 2\,x2\,x3 + x3^2)\,d1 - x2\,(x1^2 - x1\,x3 + x2^2 + 2\,x2\,x3 + x3^2)\,d2 + x1\,(2\,x1\,x2 + x1\,x3 + x2^2)\,d3 - 2\,x1\,x2 - x1\,x3 + x1^2], \quad \textbf{(11)}$$
$$[x1\,(x1 - x3)\,(2\,x2 - x1 + 2\,x3)\,d3 + (x2 + x3)^2\,(x1^2 - x1\,x3 + x2^2 + 2\,x2\,x3 + x3^2)\,d2^2 + 2\,x1\,(x1^2 - x1\,x3 + x2^2 + 2\,x2\,x3 + x3^2)\,d2 + x1\,(-2\,x2 + x1 - 2\,x3)],$$
$$[(x2 + x3)^2\,(x1^2 - x1\,x3 + x2^2 + 2\,x2\,x3 + x3^2)\,d2\,d3 + x1\,(x1^2 - x1\,x3 + x2^2 + 2\,x2\,x3 + x3^2)\,d2 + 2\,x1\,(x2 + x3)\,(x1 - x3)\,d3 - 2\,x1\,(x2 + x3)],$$
$$[(x2 + x3)^2\,(x1 - x3)\,(x1^2 - x1\,x3 + x2^2 + 2\,x2\,x3 + x3^2)\,d3^2 + (-2\,x2^4 - 8\,x3\,x2^3 - 12\,x2^2\,x3^2 - 8\,x2\,x3^3 + 2\,x1\,x3^3 + 2\,x1^3\,x2 - 3\,x1^2\,x3^2 - 4\,x1^2\,x2\,x3 - 2\,x3^4 + x1^4$$

$$+ 2\,x1\,x2\,x3^2)\,d3 - 2\,x1\,x2^2 - 2\,x2\,x1^2 - 2\,x1\,x2\,x3 - x1^3 - x3\,x1^2]]$$

We write the system as an integrable connection:

```
> C:=OreModules[Connection](R,Alg):
```

If we want to know the basis of the associated module in which the connection is written, we can use the procedure KBasis of OreModules:

```
> OreModules[KBasis](R,Alg);
```

$$[\lambda_1, d3\,\lambda_1, d2\,\lambda_1] \tag{12}$$

We use our procedure for computing hyperexponential solutions of the integrable connection:

```
> HyperexpSols:=HyperexponentialSolutions(C,[x1,x2,x3]);
```

$$HyperexpSols := \tag{13}$$

$$\begin{bmatrix} x1\,x2\,\mathrm{e}^{\frac{x1}{x2+x3}} & -\mathrm{e}^{\frac{x1}{x2+x3}} & \frac{1}{x1-x3} \\ -\frac{\mathrm{e}^{\frac{x1}{x2+x3}}\,x1^2\,x2}{(x2+x3)^2} & \frac{\mathrm{e}^{\frac{x1}{x2+x3}}\,x1}{(x2+x3)^2} & \frac{1}{(x1-x3)^2} \\ -\frac{(x1\,x2 - x2^2 - 2\,x2\,x3 - x3^2)\,\mathrm{e}^{\frac{x1}{x2+x3}}\,x1}{(x2+x3)^2} & \frac{\mathrm{e}^{\frac{x1}{x2+x3}}\,x1}{(x2+x3)^2} & 0 \end{bmatrix}$$

The system admits one non-zero rational solution, namely 1/(x1-x3) and two other non-zero hyperexponential solutions given by exp(x1/(x2+x3)) and x1*x2*exp(x1/(x2+x3)).

Consider Example 4.1 in *Z. Li, F. Schwarz, and S. Tsarev.Factoring systems of pde's with finite-dimensional solution space,*
Journal of Symbolic Computation, 36:443--471, 2003.

We define the OreAlgebra (needed for OreModules):

```
> Alg:=DefineOreAlgebra(diff=[d1,x1],diff=[d2,x2],diff=[d3,x3],
  polynom=[x1,x2,x3],comm=[]):
```

We give the equations of the system (be careful there is a typo in the fifth equation in the paper):

```
> R:=matrix(6,2,[
  (x1*x3-x1*x2)+(x1*x2-1)*d2+(1-x1*x3)*d3,0,(x2*x3-x1*x2)+(x1*x2-1)*
  d1+(1-x2*x3)*d3,0,-x2*x3,d1-x2*x3,-x1*x3,d2-x1*x3,d3^2-(x1*x2+1)*
  d3+x1*x2,0,-x1*x2,d3-x1*x2
  ]);
```

$$R := \begin{bmatrix} x1\,x3 - x1\,x2 + (-1 + x1\,x2)\,d2 + (1 - x1\,x3)\,d3 & 0 \\ x2\,x3 - x1\,x2 + (-1 + x1\,x2)\,d1 + (1 - x2\,x3)\,d3 & 0 \\ -x2\,x3 & d1 - x2\,x3 \\ -x1\,x3 & d2 - x1\,x3 \\ d3^2 - (x1\,x2 + 1)\,d3 + x1\,x2 & 0 \\ -x1\,x2 & d3 - x1\,x2 \end{bmatrix} \tag{14}$$

We write the system as an integrable connection:

```
> C:=OreModules[Connection](R,Alg);
```

$$C := \left[\begin{bmatrix} x2\,x3 & 0 \\ x2\,x3 & x2\,x3 \end{bmatrix}, \begin{bmatrix} x1\,x3 & 0 \\ x1\,x3 & x1\,x3 \end{bmatrix}, \begin{bmatrix} x1\,x2 & 0 \\ x1\,x2 & x1\,x2 \end{bmatrix} \right] \qquad \textbf{(15)}$$

If we want to know the basis of the associated module in which the connection is written, we can use the procedure KBasis of OreModules:

```
> OreModules[KBasis](R,Alg);
```

$$[\lambda_1, \lambda_2] \qquad \textbf{(16)}$$

We use our procedure for computing hyperexponential solutions of the integrable connection:

```
> HyperexpSols:=HyperexponentialSolutions(C,[x1,x2,x3]);
```

$$HyperexpSols := \begin{bmatrix} e^{x1\,x2\,x3} & 0 \\ e^{x1\,x2\,x3}\,x1\,x2\,x3 & e^{x1\,x2\,x3} \end{bmatrix} \qquad \textbf{(17)}$$

The system admits two non-trivial hyperexponential solutions, namely (0,exp(x1*x2*x3))^T and (exp(x1*x2*x3),x1*x2*x3*exp(x1*x2*x3))^T.